# Maintainer: Alexey Pavlov <alexpux@gmail.com>

_realname=Archive-Zip
pkgname=perl-${_realname}
pkgver=1.60
pkgrel=1
pkgdesc="Provide a perl interface to ZIP archive files"
arch=('any')
license=('GPL' 'PerlArtistic')
url="http://search.cpan.org/dist/Archive-Zip/"
depends=('perl')
groups=('perl-modules')
options=('!emptydirs')
source=(https://www.cpan.org/authors/id/P/PH/PHRED/${_realname}-${pkgver}.tar.gz)
sha256sums=('eac75b05f308e860aa860c3094aa4e7915d3d31080e953e49bc9c38130f5c20b')

build() {
  cd ${_realname}-${pkgver}
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd ${_realname}-${pkgver}
  make test
}

package() {
  cd ${_realname}-${pkgver}
  make DESTDIR="${pkgdir}" install
}
